
Hierarchical models
Data
A study was made to investigate how different factors affect the response variable y. Two medium-scale soil preparation techniques are compared. Each technique was used at 8 sites and at each site 3 plots were randomly selected. Measurements were made at 3 trees within each plot. The data set contains the following variables:

treat 
Treatment, soil preparation (1, 2)
site 
8 sites for each treatment (1,…,8)

plot 
3 plots selected within each site (1,…,3)

tree 
3 measurements made within each plot (= 3 trees selected randomly within each plot)

 y 
Response variable
Question 

Compare the two treatments, accounting at the same time for the sampling structure of the data. 
data forest;

  input treat site plot tree y @@;

  datalines;

1 1 1 1 2006 1 1 1 2 1999 1 1 1 3 2007 1 1 2 1 1980

1 1 2 2 1988 1 1 2 3 1982 1 1 3 1 2000 1 1 3 2 1998

1 1 3 3 2007 1 2 1 1 1991 1 2 1 2 1990 1 2 1 3 1988

1 2 2 1 1987 1 2 2 2 1989 1 2 2 3 1988 1 2 3 1 1985

1 2 3 2 1983 1 2 3 3 1989 1 3 1 1 2000 1 3 1 2 2004

1 3 1 3 2004 1 3 2 1 2001 1 3 2 2 1996 1 3 2 3 2004

1 3 3 1 1999 1 3 3 2 2000 1 3 3 3 2002 1 4 1 1 1997

1 4 1 2 1994 1 4 1 3 1996 1 4 2 1 1996 1 4 2 2 2000

1 4 2 3 2002 1 4 3 1 1987 1 4 3 2 1990 1 4 3 3 1995

2 5 1 1 2013 2 5 1 2 2004 2 5 1 3 2009 2 5 2 1 2023

2 5 2 2 2018 2 5 2 3 2010 2 5 3 1 2020 2 5 3 2 2023

2 5 3 3 2015 2 6 1 1 2032 2 6 1 2 2036 2 6 1 3 2030

2 6 2 1 2018 2 6 2 2 2022 2 6 2 3 2026 2 6 3 1 2009

2 6 3 2 2010 2 6 3 3 2011 2 7 1 1 1984 2 7 1 2 1993

2 7 1 3 1993 2 7 2 1 1992 2 7 2 2 1992 2 7 2 3 1990

2 7 3 1 1996 2 7 3 2 1993 2 7 3 3 1987 2 8 1 1 1996

2 8 1 2 1989 2 8 1 3 1996 2 8 2 1 1997 2 8 2 2 1993

2 8 2 3 1996 2 8 3 1 1990 2 8 3 2 1989 2 8 3 3 1992

; run; 

Solution: 

Here we compare two levels of the fixed factor treatment (‘treat’). Data are collected by a hierarchical design including sites within treatment, plots within sites and trees within plots. 
The model 

yij=(+(i + bij + cijk+ eijkl
(i 
Fixed effect of treatment i

bij
Random effect of site j within treatment i
cijk
Random effect of plot k within site j (and treatment i)
eijkl
Random residual
In SAS: 

The program can be written as: 

proc mixed data=forest;

class treat site plot;

model y= treat /ddfm=kr;

random site(treat) plot(site*treat);

run;

Note: 

random site(treat) plot(site*treat); 

can also be written as

random site*treat plot*site*treat;

ddfm=kr is an adjustment for the number of degrees of freedom in the denominator using the Kenward-Rogers method. Adjustment can be advantageous if the design is unbalanced or if variances are unequal.
The comparison of the two treatments is made by F-test. No significant differences between the treatments. 
	Type 3 Tests of Fixed Effects

	Effect
	Num DF
	Den DF
	F Value
	Pr > F

	treat
	1
	6
	1.53
	0.2629


Variance components are estimated by the model. We see that a lot of variation is contributed by the different sites. 

	Covariance Parameter Estimates

	Cov Parm
	Estimate

	site(treat)
	119.89

	plot(treat*site)
	35.8657

	Residual
	12.5694


We can use the covariance parameter estimates to quantify the sources of random variation in the model. The variation of a treatment mean value consists variation between sites, between plots within site and between trees within plot (residualvariation). If we look at the percentage contribution of each source (e.g. by the plot below), we see that the main source of variation is site, followed by plot. In a new experiment we would select more sites and maybe less trees.

Residual check: 

Depending on your installation of SAS you get plots directly in your Viewer window or you need to request them. To see all plots write plots=(all) in the model statement.
ods graphics on;

proc mixed data=forest plots=(all);

class treat site plot tree;

model y= treat/  solution;

random site(treat) plot(site*treat);

run;

ods graphics off;

In mixed models we check if the residuals are normally distributed and if the variance is equal in all groups (plot: conditional residual plots -> Residuals). No problems are detected here. 

To compare the two treatments we use the lsmeans and pdiff statements. Of course, since we only have two treatments and the F-test is not significant, in this case we already know that there will be no significant difference between the treatments. The p-value is the same as in the F-test. 

ods graphics on;

proc mixed data=forest plots=(all);

class treat site plot tree;

model y= treat/  solution;

random site(treat) plot(site*treat);

lsmeans treat /pdiff;

run;

ods graphics off;
